

Bài 7: KẾT NỐI DATABASE.

1. ADO là gì?

ADO Có thể dùng để truy cập DL từ trang web.

- ADO là công nghệ của Microsoft
- ADO được viết tắt là **ActiveX Data Objects**
- ADO là một thành phần của Active-X
- ADO tự động cài đặt khi cài IIS
- ADO là môi trường lập trình giao tiếp với CSDL- database

1. Cách truy xuất database bằng ADO trong ASP.

Cách thông thường để truy cập CSDL từ trang web ASP như sau:

1. Create an ADO connection to a database
2. Open the database connection
3. Create an ADO recordset
4. Open the recordset
5. Extract the data you need from the recordset
6. Close the recordset
7. Close the connection

2. Tạo kết nối ADO (ADO Connection).

Trước khi truy cập dữ liệu, bạn phải thiết lập kết nối đến CSDL đó. Có 02 cách để thiết lập kết nối.

a/ Kết nối trực tiếp(DSN-less Database Connection)

Đây là cách đơn giản nhất thường kết nối với Microsoft Access database.

Giả sử bạn có một Database có tên "northwind.mdb" đặt tại thư mục "c:/webdata/", bạn có thể kết nối đến CSDL như sau:

```
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Provider="Microsoft.Jet.OLEDB.4.0"  
conn.Open "c:/webdata/northwind.mdb"  
%>
```

b/ Kết nối thông qua ODBC(ODBC Database Connection)

Nếu bạn có ODBC database có tên là "**northwind**" bạn có thể kết nối vào CSDL này như sau:

```
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Open "northwind"  
%>
```

Với kết nối ODBC connection, Bạn có thể kết nối đến bất kỳ CSDL nào và ở bất kỳ máy nào trong hệ thống mạng của bạn, miễn là ODBC connection tồn tại.

Cách tạo ODBC Connection với MS Access Database

1. Open the **ODBC** icon in your Control Panel.
2. Choose the **System DSN** tab.
3. Click on **Add** in the System DSN tab.
4. **Select** the Microsoft Access Driver. Click **Finish**.
5. In the next screen, click **Select** to locate the database.
6. Give the database a **Data Source Name** (DSN).
7. Click **OK**.

3. Sử dụng ADO Recordset

Để có thể đọc dữ liệu từ database, đầu tiên dữ liệu phải được nạp vào recordset.

Cách tạo ADO Recordset

Sau khi ADO Database Connection đã được tạo, tiếp theo ta phải tạo ADO Recordset.

Giả sử ta có database có tên "Northwind", Chúng ta có thể truy cập dữ liệu từ bản "Customers" như sau:

```
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Provider="Microsoft.Jet.OLEDB.4.0"  
conn.Open "c:/webdata/northwind.mdb"  
set rs=Server.CreateObject("ADODB.recordset")  
rs.Open "Customers", conn  
%>
```

Cách tạo ADO SQL Recordset

Chúng ta có thể truy cập dữ liệu từ bản "Customers" sử dụng câu lệnh SQL:

```
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Provider="Microsoft.Jet.OLEDB.4.0"  
conn.Open "c:/webdata/northwind.mdb"  
set rs=Server.CreateObject("ADODB.recordset")  
rs.Open "Select * from Customers", conn  
%>
```

Lấy dữ liệu từ Recordset

Sau khi recordset được mở, chúng ta có thể lấy dữ liệu recordset.

Chúng ta có thể truy xuất dữ liệu từ bản **Customers** trong CSDL **Northwind** như sau :

```
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Provider="Microsoft.Jet.OLEDB.4.0"  
conn.Open "c:/webdata/northwind.mdb"  
set rs=Server.CreateObject("ADODB.recordset")  
rs.Open "Select * from Customers", conn  
for each x in rs.fields  
    response.write(x.name)  
    response.write(" = ")  
    response.write(x.value)  
next  
%>
```

4. Một số ví dụ.

Hiển thị tên Field và giá trị của Field

Chúng ta có thể truy xuất dữ liệu từ bản **Customers** trong CSDL **Northwind** như sau :

```
<html>  
<body>  
<%  
set conn=Server.CreateObject("ADODB.Connection")  
conn.Provider="Microsoft.Jet.OLEDB.4.0"  
conn.Open "c:/webdata/northwind.mdb"  
set rs = Server.CreateObject("ADODB.recordset")  
rs.Open "SELECT * FROM Customers", conn  
do until rs.EOF  
    for each x in rs.Fields  
        Response.Write(x.name)
```

```
        Response.Write(" = ")
        Response.Write(x.value & "<br />")
    next
    Response.Write("<br />")
    rs.MoveNext
loop
rs.close
conn.close
%>
</body>
</html>
```

Kết quả

```
CustomerID = ALFKI
CompanyName = Alfreds Futterkiste
ContactName = Maria Anders
ContactTitle = Sales Representative
Address = Obere Str. 57
City = Berlin
PostalCode = 12209
Country = Germany

CustomerID = BERGS
CompanyName = Berglunds snabbköp
ContactName = Christina Berglund
ContactTitle = Order Administrator
Address = Berguvsvägen 8
City = Luleå
PostalCode = S-958 22
Country = Sweden

CustomerID = CENTC
CompanyName = Centro comercial Moctezuma
ContactName = Francisco Chang
ContactTitle = Marketing Manager
Address = Sierras de Granada 9993
City = México D.F.
PostalCode = 05022
Country = Mexico

....
....
....
```

Hiển thị tên Field và giá trị của Field trong HTML table

```
<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
set rs = Server.CreateObject("ADODB.recordset")
rs.Open "SELECT Companyname, Contactname FROM Customers", conn
%>
<table border="1" width="100%">
<%do until rs.EOF%>
    <tr>
    <%for each x in rs.Fields%>
        <td><%Response.Write(x.value)%></td>
    <%next
    rs.MoveNext%>
    </tr>
<%loop
rs.close
conn.close
%>
</table>
</body>
</html>
```

Kết quả:

Alfreds Futterkiste	Maria Anders
Berglunds snabbköp	Christina Berglund
Centro comercial Moctezuma	Francisco Chang
Ernst Handel	Roland Mendel
FISSA Fabrica Inter. Salchichas S.A.	Diego Roel
Galería del gastrónomo	Eduardo Saavedra
Island Trading	Helen Bennett
Königlich Essen	Philip Cramer
Laughing Bacchus Wine Cellars	Yoshi Tannamuri
Magazzini Alimentari Riuniti	Giovanni Rovelli
North/South	Simon Crowther
Paris spécialités	Marie Bertrand
Rattlesnake Canyon Grocery	Paula Wilson
Simons bistro	Jytte Petersen

The Big Cheese	Liz Nixon
Vaffeljernet	Palle Ibsen
Wolski Zajazd	Zbyszek Piestrzeniewicz

Thêm Headers vào Table

```
<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
set rs = Server.CreateObject("ADODB.recordset")
sql="SELECT Companyname, Contactname FROM Customers"
rs.Open sql, conn
%>
<table border="1" width="100%">
  <tr>
    <%for each x in rs.Fields
      response.write("<th>" & x.name & "</th>")
    next%>
  </tr>
  <%do until rs.EOF%>
    <tr>
      <%for each x in rs.Fields%>
        <td><%Response.Write(x.value)%></td>
      <%next
      rs.MoveNext%>
    </tr>
  <%loop
  rs.close
  conn.close
  %>
</table>
</body>
</html>
```

Kết quả:

Companyname	Contactname
Alfreds Futterkiste	Maria Anders
Berglunds snabbköp	Christina Berglund
Centro comercial Moctezuma	Francisco Chang
Ernst Handel	Roland Mendel

FISSA Fabrica Inter. Salchichas S.A.	Diego Roel
Galería del gastrónomo	Eduardo Saavedra
Island Trading	Helen Bennett
Königlich Essen	Philip Cramer
Laughing Bacchus Wine Cellars	Yoshi Tannamuri
Magazzini Alimentari Riuniti	Giovanni Rovelli
North/South	Simon Crowther
Paris spécialités	Marie Bertrand
Rattlesnake Canyon Grocery	Paula Wilson
Simons bistro	Jytte Petersen
The Big Cheese	Liz Nixon
Vaffeljernet	Palle Ibsen
Wolski Zajazd	Zbyszek Piestrzeniewicz

5. QUERY

Display Selected Data

Chúng ta chỉ hiển thị những bản ghi từ bản "Customers" mà trường "Companyname" bắt đầu bằng chữ A

```
<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
set rs=Server.CreateObject("ADODB.recordset")
sql="SELECT Companyname, Contactname FROM Customers
WHERE CompanyName LIKE 'A%'"
rs.Open sql, conn
%>
<table border="1" width="100%">
  <tr>
    <%for each x in rs.Fields
      response.write("<th>" & x.name & "</th>")
    next%>
  </tr>
  <%do until rs.EOF%>
    <tr>
      <%for each x in rs.Fields%>
        <td><%Response.Write(x.value)%></td>
      <%next
```

```

        rs.MoveNext%>
    </tr>
<%loop
rs.close
conn.close%>
</table>
</body>
</html>

```

Sort the Data

Hiển thị "Companyname" và "Contactname" từ bản "Customers", sắp xếp bởi "Companyname":

```

<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
set rs = Server.CreateObject("ADODB.recordset")
sql="SELECT Companyname, Contactname FROM
Customers ORDER BY CompanyName"
rs.Open sql, conn
%>

<table border="1" width="100%">
  <tr>
    <%for each x in rs.Fields
      response.write("<th>" & x.name & "</th>")
    next%>
  </tr>
  <%do until rs.EOF%>
    <tr>
      <%for each x in rs.Fields%>
        <td><%Response.Write(x.value)%></td>
      <%next
      rs.MoveNext%>
    </tr>
  <%loop
  rs.close
  conn.close%>
</table>
</body>

</html>

```


Add a Record to a Table in a Database

Chúng ta có thể thêm 1 bản ghi mới vào CSDL Northwind. Trước tiên bạn phải tạo ra 01 form nhập liệu:

```
<html>
<body>
<form method="post" action="demo_add.asp">
<table>
<tr>
<td>CustomerID:</td>
<td><input name="custid"></td>
</tr><tr>
<td>Company Name:</td>
<td><input name="compname"></td>
</tr><tr>
<td>Contact Name:</td>
<td><input name="contname"></td>
</tr><tr>
<td>Address:</td>
<td><input name="address"></td>
</tr><tr>
<td>City:</td>
<td><input name="city"></td>
</tr><tr>
<td>Postal Code:</td>
<td><input name="postcode"></td>
</tr><tr>
<td>Country:</td>
<td><input name="country"></td>
</tr>
</table>
<br /><br />
<input type="submit" value="Add New">
<input type="reset" value="Cancel">
</form>
</body>
</html>
```

Khi bạn bấm vào nút **submit**(add New) sẽ chuyển sang file "demo_add.asp".
"demo_add.asp" chứa đoạn mã thêm 01 record mới vào bản Customers như sau:

```
<html>
<body>
<%
```

```

set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
sql="INSERT INTO customers (customerID,companyname,"
sql=sql & "contactname,address,city,postalcode,country)"
sql=sql & " VALUES "
sql=sql & "(" & Request.Form("custid") & ","
sql=sql & Request.Form("compname") & ","
sql=sql & Request.Form("contname") & ","
sql=sql & Request.Form("address") & ","
sql=sql & Request.Form("city") & ","
sql=sql & Request.Form("postcode") & ","
sql=sql & Request.Form("country") & ")")
on error resume next
conn.Execute sql,recaffected
if err<>0 then
    Response.Write("No update permissions!")
else
    Response.Write("<h3>" & recaffected & " record added</h3>")
end if
conn.close
%>

</body>
</html>

```

Update a Record in a Table

Bạn muốn cập nhật lại thông tin 1 record trong bản Customers trong CSDL: Northwind. Đầu tiên ta hiển thị tất cả các records trong bảng Customers, mỗi record lấy customerID làm tham số :

```

<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
set rs=Server.CreateObject("ADODB.Recordset")
rs.open "SELECT * FROM customers",conn
%>
<h2>List Database</h2>
<table border="1" width="100%">
<tr>
<%
for each x in rs.Fields

```

```

        response.write("<th>" & ucase(x.name) & "</th>")
next
%>
</tr>
<% do until rs.EOF %>
<tr>
<form method="post" action="demo_update.asp">
<%
for each x in rs.Fields
    if lcase(x.name)="customerid" then%>
        <td>
            <input type="submit" name="customerID" value="<%=x.value%>">
        </td>
    <%else%>
        <td><%Response.Write(x.value)%></td>
    <%end if
next
%>
</form>
<%rs.MoveNext%>
</tr>
<%
loop
conn.close
%>
</table>
</body>
</html>

```

Khi bạn click vào "customerID" sẽ gọi đến file "demo_update.asp". tập tin "demo_update.asp" file này chứa đoạn mã tạo ra form chứa các field từ customerID đã. Form này có 01 nút lệnh "Update record" sẽ thực hiện việc lưu dữ liệu:

```

<html>
<body>
<h2>Update Record</h2>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
cid=Request.Form("customerID")
if Request.form("companyname")="" then
    set rs=Server.CreateObject("ADODB.Recordset")
    rs.open "SELECT * FROM customers WHERE customerID='" & cid & "
%>

```

```

<form method="post" action="demo_update.asp">
<table>
<%for each x in rs.Fields%>
<tr>
<td><%=x.name%></td>
<td><input name="<%=x.name%>" value="<%=x.value%>"></td>
<%next%>
</tr>
</table>
<br /><br />
<input type="submit" value="Update record">
</form>
<%
else
    sql="UPDATE customers SET "
    sql=sql & "companyname='" & Request.Form("companyname") & "',"
    sql=sql & "contactname='" & Request.Form("contactname") & "',"
    sql=sql & "address='" & Request.Form("address") & "',"
    sql=sql & "city='" & Request.Form("city") & "',"
    sql=sql & "postalcode='" & Request.Form("postalcode") & "',"
    sql=sql & "country='" & Request.Form("country") & "'"
    sql=sql & " WHERE customerID='" & cid & "'"
    on error resume next
    conn.Execute sql
    if err<>0 then
        response.write("No update permissions!")
    else
        response.write("Record " & cid & " was updated!")
    end if
end if
conn.close
%>
</body>
</html>

```

Delete a Record in a Table

Bạn muốn xóa 1 record trong bảng Customers trong CSDL Northwind. Trước tiên bạn phải hiển thị ra tất cả các record trong bảng đó để chọn xóa:

```

<html>
<body>
<%
set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"

```

```

set rs=Server.CreateObject("ADODB.Recordset")
rs.open "SELECT * FROM customers",conn
%>
<h2>List Database</h2>
<table border="1" width="100%">
<tr>
<%
for each x in rs.Fields
    response.write("<th>" & ucase(x.name) & "</th>")
next
%>
</tr>
<% do until rs.EOF %>
<tr>
<form method="post" action="demo_delete.asp">
<%
for each x in rs.Fields
    if x.name="customerID" then%>
        <td>
            <input type="submit" name="customerID" value="<%=x.value%>">
        </td>
    <%else%>
        <td><%Response.Write(x.value)%></td>
    <%end if
next
%>
</form>
<%rs.MoveNext%>
</tr>
<%
loop
conn.close
%>
</table>
</body>
</html>

```

Nếu bạn click vào cột "customerID" một file mới được gọi đến "demo_delete.asp". File "demo_delete.asp" chứa đoạn mã hiển thị dữ liệu bạn muốn xóa. Tiếp đó có nút lệnh "Delete record" để xóa record mà bạn chọn:

```

<html>
<body>
<h2>Delete Record</h2>
<%

```

```

set conn=Server.CreateObject("ADODB.Connection")
conn.Provider="Microsoft.Jet.OLEDB.4.0"
conn.Open "c:/webdata/northwind.mdb"
cid=Request.Form("customerID")
if Request.form("companyname")="" then
    set rs=Server.CreateObject("ADODB.Recordset")
    rs.open "SELECT * FROM customers WHERE customerID='" & cid & "'
    %>
    <form method="post" action="demo_delete.asp">
    <table>
    <%for each x in rs.Fields%>
    <tr>
    <td><%=x.name%></td>
    <td><input name="<%=x.name%>" value="<%=x.value%>"></td>
    <%next%>
    </tr>
    </table>
    <br /><br />
    <input type="submit" value="Delete record">
    </form>
<%
else
    sql="DELETE FROM customers"
    sql=sql & " WHERE customerID='" & cid & "'"
    on error resume next
    conn.Execute sql
    if err<>0 then
        response.write("No update permissions!")
    else
        response.write("Record " & cid & " was deleted!")
    end if
end if
conn.close
%>
</body>
</html>

```